



LE POMPISTE

Dans les programmes

Mettre un problème en équation. Travail sur les fonctions polynômes de degré 2.

Un pompiste vend le litre d'essence au prix de 1,20€. Le prix d'achat est pour lui de 0,85€ le litre. Il sait qu'il peut compter sur une vente journalière de 1 000 litres et qu'à chaque baisse de 1 centime qu'il consent pour le prix du litre, il vendra 100 litres de plus par jour.

On note x le nombre de baisses de 1 centime. On note $f(x)$ le bénéfice (en euros) correspondant.

1. A l'aide d'un programme, déterminer le prix de vente du carburant correspondant à un bénéfice maximal.
2. Si cela n'a pas été fait dans l'algorithme précédent, quel algorithme peut-on écrire pour vérifier l'unicité de la réponse au problème précédent ?
3. En utilisant les résultats de la question précédente, que pouvez vous dire du maximum de la fonction f définie sur \mathbb{R} par :

$$f(x) = (1000 + 100x)(1,20 - x/100 - 0,85) \quad ?$$

Éléments de réponses – XCAS

1. Recherche d'une solution.

Pour x baisses, le nombre de litres vendus est de $1000 + 100x$ avec un bénéfice de $1,20 - \frac{x}{100} - 0,85$ € par litre.

Le nombre maximal de baisses est de 35 (écart entre 1,20 et 0,85).

Xcas

```
nb_baisses:=0;;
benef_max:=1000*(1.20-0.85) ;;
pour k de 1 jusque 35 faire
benef:=(1000+100*k)*(1.20-k/100-0.85);
si benef>benef_max alors benef_max:=benef; nb_baisses:=k; fsi ;
fpour ;;
print(nb_baisses) ;;
```

Le programme annonce un nombre optimal de baisses de 12, soit un prix de vente de 1,08 €.

2. Problème de l'unicité

On peut faire une recherche des éventuelles autres valeurs du nombre de baisses correspondant à ce bénéfice maximal comme suit :

Xcas

```
nb_baisses:=0;;
benef_max:=1000*(1.20-0.85) ;;
pour k de 1 jusque 35 faire
benef:=(1000+100*k)*(1.20-k/100-0.85);
si benef>benef_max alors benef_max:=benef; fsi ;
fpour ;;
pour k de 0 jusque 35 faire
benef:=(1000+100*k)*(1.20-k/100-0.85);
si benef==benef_max alors print(k); fsi ;
fpour ;;
```

Avec 12 ou 13 baisses, soit avec des prix du litre de 1,08 € ou 1,07 €, le bénéfice est maximal (bénéfice de 506 € dans les deux cas).

3. Traitement graphique.

On peut ajouter un affichage des points de coordonnées (nombre de baisses ; bénéfice correspondant) :

Xcas


```
nb_baisses:=0;;
benef_max:=1000*(1.20-0.85) ;;

pour k de 1 jusque 35 faire
    benef:=(1000+100*k)*(1.20-k/100-0.85);
    si benef>benef_max alors benef_max:=benef; fsi;
fpour ;;

liste_max:=[];;
liste_autre:=[];;
pour k de 0 jusque 35 faire
    benef:=(1000+100*k)*(1.20-k/100-0.85);
    si benef==benef_max
        alors afficher(k);liste_max:=append(liste_max , point(k, benef));
    sinon liste_autre:=append(liste_autre , point(k, benef));
    fsi;
fpour ;;

affichage(liste_max , point_width_4+red);
affichage(liste_autre , point_width_4+green);
```

Cliquer sur la case "auto" à droite de l'écran graphique qui s'affiche après exécution pour visualiser les points.



4. Maximum de la fonction.

Dans une classe où l'on aura déjà travaillé sur la symétrie des courbes représentant les fonctions du second degré, on conjecturera que l'abscisse du maximum de la fonction est $\frac{12+13}{2} = 12,5$. Ce que l'on pourra ensuite vérifier par le calcul $f(x) - f(12,5) = -(x - 12,5)^2$.